

# ΥΠΟΛΟΓΙΣΤΕΣ Ι

## ΣΥΝΑΡΤΗΣΕΙΣ

## Τι είναι μια συνάρτηση;

- Μια ομάδα εντολών, σχεδιασμένη να εκτελεί έναν υπολογισμό και να γυρνάει το αποτέλεσμα
  - Ιδανικές για περιπτώσεις που ο υπολογισμός επαναλαμβάνεται πολλές φορές μέσα στο πρόγραμμα
  - Συντελούν σημαντικά στην καθαρότητα ενός προγράμματος
  - Συντελούν σημαντικά στην μεταβατικότητα ενός προγράμματος

Παράδειγμα: όλες οι «εσωτερικές» συναρτήσεις της Fortran:  
`cos, sin, acos, asin, sqrt, abs`, κτλ

## Παράδειγμα συνάρτησης: υπολογισμός τετραγώνου

- Είναι ήδη γνωστή η εσωτερική συνάρτηση για την τετραγωνική ρίζα `sqrt`.
- Πως θα γράφαμε μια συνάρτηση για το τετράγωνο;
- Επιλέγουμε όνομα που περιγράφει συνοπτικά τι θα κάνει η συνάρτηση: π.χ. `sq`
- Επιλέγουμε τον τύπο των μεταβλητών που δέχεται
  - π.χ. `double precision`
- Επιλέγουμε τον τύπο της μεταβλητής που επιστρέφει
  - εδώ υποχρεωτικά `double precision`

## Δήλωση συνάρτησης `sq`

```
DOUBLE PRECISION FUNCTION SQ ( X )  
  IMPLICIT NONE  
  DOUBLE PRECISION X  
  SQ = X**2  
  RETURN  
END
```

- Γίνεται έξω από το κυρίως πρόγραμμα
  - μετά το `END` του κυρίως προγράμματος
- Χρησιμοποιείται η εντολή `FUNCTION`
  - αριστερά από το `FUNCTION` ο **τύπος επιστροφής**
  - δεξιά από το `FUNCTION` το **όνομα** της συνάρτησης και η **λίστα εισόδου**
- Οι μεταβλητές εισόδου ξαναδηλώνονται
- Το **όνομα της συνάρτησης είναι η μεταβλητή εξόδου**
  - είναι ήδη δηλωμένη: είναι ο τύπος της συνάρτησης
- Η εντολή `RETURN` μας επιστρέφει στο κυρίως πρόγραμμα
  - επιστρέφει την μεταβλητή εξόδου στο κυρίως πρόγραμμα

## Γενική δήλωση συνάρτησης

```

τύπος FUNCTION όνομα (μεταβλητές εισόδου)
  IMPLICIT NONE
  δηλώσεις μεταβλητών (εισόδου και τοπικών)
  εντολές
  όνομα = ...
  RETURN
END

```

- **τύπος:** π.χ. INTEGER, DOUBLE PRECISION
- **όνομα:** κάτι που περιγράφει τον υπολογισμό που γίνεται
- **μεταβλητές εισόδου:** από 0 μέχρι και 265 (χωρισμένες με κόμματα) Απλές μεταβλητές ή/και πίνακες. Όποιου τύπου θέλουμε. Αρκεί να έχουν πάρει κάποια τιμή από το κυρίως πρόγραμμα
- **δηλώσεις μεταβλητών:** οποσδήποτε των μεταβλητών εισόδου, και όποιες τοπικής μεταβλητής δηλώσουμε μέσα στην συνάρτηση
- **εντολές:** ο κώδικας της συνάρτησης
- **όνομα=...:** καθώς το όνομα της συνάρτησης ΕΙΝΑΙ η μεταβλητή εξόδου, πρέπει να της δώσουμε τιμή πριν την έξοδο
- **RETURN:** επιστρέφει την μεταβλητή εξόδου στο κυρίως πρόγραμμα

5

## Παράδειγμα #1: Συνάρτηση για το μέτρο δισδιάστατου διανύσματος

```

DOUBLE PRECISION FUNCTION MAGNITUDE ( X, Y )
  IMPLICIT NONE
  DOUBLE PRECISION X, Y

  MAGNITUDE = SQRT (X**2 + Y**2)

  RETURN
END

```

6

## Παράδειγμα #2: Συνάρτηση για τον μεγαλύτερο τριών αριθμών

```

DOUBLE PRECISION FUNCTION MAX3 ( X, Y, Z )
  IMPLICIT NONE
  DOUBLE PRECISION X, Y, Z, M

  IF (X > Y) THEN
    M = X
  ELSE
    M = Y
  END IF

  IF (Z > M) THEN
    MAX3 = Z
  ELSE
    MAX3 = M
  END IF

  RETURN
END

```

7

## Κλήση συνάρτησης

- Καλείται όπως και οι «εσωτερικές» συναρτήσεις
- Μόνη διαφορά: επειδή το όνομα της συνάρτησης είναι και αυτό μεταβλητή, πρέπει να δηλωθεί στο πρόγραμμα που την καλεί

```

PROGRAM TEST_SQ
  IMPLICIT NONE
  DOUBLE PRECISION X, SQ
  WRITE(*,*) 'ΔΩΣΕ ΕΝΑ ΝΟΥΜΕΡΟ'
  READ(*,*) X
  WRITE(*,*) 'ΤΟ ΤΕΤΡΑΓΩΝΟ ΤΟΥ', X, 'ΕΙΝΑΙ', SQ(X)
  END

  DOUBLE PRECISION FUNCTION SQ ( X )
  IMPLICIT NONE
  DOUBLE PRECISION X
  SQ = X**2
  RETURN
END

```

8

## Κλήση συνάρτησης: λεπτομέρεια 1

- Μέσα στην συνάρτηση δεν είμαστε υποχρεωμένοι να διατηρούμε τα ίδια ονόματα για τις μεταβλητές εισόδου, αρκεί οι τύποι να είναι σωστοί, και να δίνονται με την σωστή σειρά

```
PROGRAM TEST_SQ
  IMPLICIT NONE
  DOUBLE PRECISION X, SQ
  WRITE(*,*) 'ΔΩΣΕ ΕΝΑ ΝΟΥΜΕΡΟ'
  READ(*,*) X
  WRITE(*,*) 'ΤΟ ΤΕΤΡΑΓΩΝΟ ΤΟΥ', X, 'ΕΙΝΑΙ', SQ(X)
  END

  DOUBLE PRECISION FUNCTION SQ ( Y )
  IMPLICIT NONE
  DOUBLE PRECISION Y
  SQ = Y**2
  RETURN
END
```

9

## Παράδειγμα #3: Συνάρτηση για το μέτρο δισδιάστατου διανύσματος

```
PROGRAM VECTOR2D
  IMPLICIT NONE
  DOUBLE PRECISION X1, Y1, MAGNITUDE, M
  WRITE(*,*) 'ΔΩΣΕ ΤΙΣ ΣΥΝΙΣΤΩΣΕΣ'
  READ(*,*) X1, Y1

  M = MAGNITUDE(X1, Y1)
  WRITE(*,*) 'ΤΟ ΜΕΤΡΟ ΤΩΝ', X1, Y1, 'ΕΙΝΑΙ', M
  END

  DOUBLE PRECISION FUNCTION MAGNITUDE ( X, Y )
  IMPLICIT NONE
  DOUBLE PRECISION X, Y
  MAGNITUDE = SQRT (X**2 + Y**2)
  RETURN
END
```

10

## Κλήση συνάρτησης: λεπτομέρεια 2

- Εαν η τιμή μιας μεταβλητής εισόδου αλλάξει μέσα στην συνάρτηση, αλλάζει και στο κυρίως πρόγραμμα. **Προσοχή να μην γίνει!!!**

```
PROGRAM VECTOR2D
  IMPLICIT NONE
  DOUBLE PRECISION X1, Y1, MAGNITUDE, M
  WRITE(*,*) 'ΔΩΣΕ ΤΙΣ ΣΥΝΙΣΤΩΣΕΣ'
  READ(*,*) X1, Y1
  M = MAGNITUDE(X1, Y1)
  WRITE(*,*) 'ΤΟ ΜΕΤΡΟ ΤΩΝ', X1, Y1, 'ΕΙΝΑΙ', M
  END

  DOUBLE PRECISION FUNCTION MAGNITUDE ( X, Y )
  IMPLICIT NONE
  DOUBLE PRECISION X, Y
  MAGNITUDE = SQRT (X**2 + Y**2)
  X = 10 * X
  RETURN
END
```

11

## Κλήση συνάρτησης: λεπτομέρεια 3

- Μπορούμε να δηλώσουμε νέες τοπικές μεταβλητές στην συνάρτηση. Μετά την επιστροφή χάνονται, δεν επηρεάζουν το κυρίως πρόγραμμα

```
PROGRAM VECTOR2D
  IMPLICIT NONE
  DOUBLE PRECISION X1, Y1, MAGNITUDE, M
  WRITE(*,*) 'ΔΩΣΕ ΤΙΣ ΣΥΝΙΣΤΩΣΕΣ'
  READ(*,*) X1, Y1
  M = MAGNITUDE(X1, Y1)
  WRITE(*,*) 'ΤΟ ΜΕΤΡΟ ΤΩΝ', X1, Y1, 'ΕΙΝΑΙ', M
  END

  DOUBLE PRECISION FUNCTION MAGNITUDE ( X, Y )
  IMPLICIT NONE
  DOUBLE PRECISION X, Y, K
  K = SQRT (X**2 + Y**2)
  MAGNITUDE = K
  RETURN
END
```

12

## Παράδειγμα #4: Συνάρτηση για τον υπολογισμό τιμής έκφρασης

Γράψτε συνάρτηση που να γυρνάει την τιμή της έκφρασης:

$$f(x) = \frac{2x^2 + 3x + 4}{5x^2 + 6x + 7}$$

```
DOUBLE PRECISION FUNCTION FUNC1 ( X )
  IMPLICIT NONE
  DOUBLE PRECISION X, Y

  Y = 5*X**2 + 6*X + 7
  IF (Y /= 0) THEN
    FUNC1 = (2*X**2 + 3*X + 4) / Y
  ELSE
    FUNC1 = 0
  END IF

  RETURN
END
```

13

## Παράδειγμα #5: Συνάρτηση για τον υπολογισμό παραγώγου έκφρασης

Γράψτε συνάρτηση που να γυρνάει την τιμή της παραγώγου της παρακάτω συνάρτησης στο  $x=x_0$

$$f(x) = \frac{2x^2 + 3x + 4}{5x^2 + 6x + 7}$$

- Υπάρχουν 2 τρόποι να επιλυθεί
  - Βρίσκουμε αναλυτικά την παράγωγο και παίρνουμε την τιμή της στο  $x=x_0$ .
  - Υπολογίζουμε αριθμητικά την τιμή της παραγώγου στο  $x=x_0$  χρησιμοποιώντας τον ορισμό της παραγώγου:

$$\left. \frac{df(x)}{dx} \right|_{x=x_0} = \lim_{e \rightarrow 0} \frac{f(x+e) - f(x-e)}{2e}$$

Τι να επιλέξουμε για  $e$ ;

14

## Παράδειγμα #5: Συνάρτηση για τον υπολογισμό παραγώγου έκφρασης

- Επιλέγουμε  $e$  πολύ μικρό σχετικά με το  $x_0$ , ας πούμε  $e=x_0/1.000.000$
- Χρησιμοποιούμε την συνάρτηση **FUNC1** που γράψαμε

```
DOUBLE PRECISION FUNCTION DERIV1 ( X )
  IMPLICIT NONE
  DOUBLE PRECISION X, E, FUNC1

  E = X * 1.0D-6
  IF (X == 0) E = 1.0D-6

  DERIV1 = (FUNC1(X+E) - FUNC1(X-E)) / (2*E)

  RETURN
END
```



15

```
PROGRAM FUNC_DERIV
  IMPLICIT NONE
  DOUBLE PRECISION X, FUNC1, DERIV1
  INTEGER I

  DO I = 1, 1000
    X = -20 + 40 * 0.001 * I
    WRITE(*,*) X, FUNC1(X), DERIV1(X)
  END DO

  END
```

$$f(x) = \frac{2x^2 + 3x + 4}{5x^2 + 6x + 7}$$




## Παράδειγμα #6: Συνάρτηση για τον υπολογισμό ολοκληρώματος

ΥΠΟΛΟΓΙΣΤΕΣ I - ΣΥΝΑΡΤΗΣΕΙΣ

- Το ολοκλήρωμα υπολογίζεται με τον κανόνα του ορθογωνίου

$$\int_{x_1}^{x_2} f(x) dx = \sum_{i=0}^N f(x_i) \Delta x$$



17

## Παράδειγμα #6: Συνάρτηση για τον υπολογισμό ολοκληρώματος έκφρασης

Γράψτε συνάρτηση που να γυρνάει την τιμή του ολοκληρώματος της παραδίπλα συνάρτησης από  $x_1$  έως  $x_2$  με 10000 διαστήματα

$$f(x) = \frac{2x^2 + 3x + 4}{5x^2 + 6x + 7}$$

ΥΠΟΛΟΓΙΣΤΕΣ I - ΣΥΝΑΡΤΗΣΕΙΣ

```
DOUBLE PRECISION FUNCTION INTEG1 ( X1, X2 )
  IMPLICIT NONE
  INTEGER I, N
  DOUBLE PRECISION X1, X2, X, DX, FUNC1


  N = 10000
  DX = (X2 - X1) / (N+1)
  INTEG1 = 0
  DO I = 0, N
    X = I * DX + DX/2 + X1
    INTEG1 = INTEG1 + FUNC1(X) * DX
  END DO
  RETURN
END
```

18

## Παράδειγμα #7: Σημείο μέσα σε κύκλο

ΥΠΟΛΟΓΙΣΤΕΣ I - ΣΥΝΑΡΤΗΣΕΙΣ

Έστω ότι δίνεται σημεία στο επίπεδο και ένας κύκλος. Δημιουργήστε την συνάρτηση `INSIDE_CIRCLE` που να υπολογίζει εάν ένα σημείο είναι μέσα ή έξω από τον κύκλο



19

## Παράδειγμα #7: Συνάρτηση λήψης απόφασης εάν σημείο είναι μέσα ή έξω από κύκλο

Είσοδος: σημείο, κέντρο κύκλου, ακτίνα κύκλου  
Έξοδος: 1 εάν είναι μέσα, 0 εάν όχι

ΥΠΟΛΟΓΙΣΤΕΣ I - ΣΥΝΑΡΤΗΣΕΙΣ

```
INTEGER FUNCTION INSIDE_CIRCLE (X, Y, X0, Y0, R)
  IMPLICIT NONE
  DOUBLE PRECISION X, Y, X0, Y0, R, D

  D = SQRT( (X-X0)**2 + (Y-Y0)**2 )
  IF (D <= R) THEN
    INSIDE_CIRCLE = 1
  ELSE
    INSIDE_CIRCLE = 0
  END IF


  RETURN
END
```

20

## Παράδειγμα #8: Αριθμός σημείων μέσα στο υπόλοιπο της τομής 3 κύκλων

ΥΠΟΛΟΓΙΣΤΕΣ I - ΣΥΝΑΡΤΗΣΕΙΣ

Έστω ότι δίνονται N σημεία στο επίπεδο. Βρείτε πόσα από αυτά βρίσκονται μέσα σε κάποιον από τους 3 κύκλους, αλλά όχι μέσα σε κοινή τομή τους (δηλαδή πόσα είναι μέσα στην ανοιχτή γαλάζια περιοχή)



## Παράδειγμα #8: Αριθμός σημείων μέσα στο υπόλοιπο της τομής 3 κύκλων (1/2)

ΥΠΟΛΟΓΙΣΤΕΣ I - ΣΥΝΑΡΤΗΣΕΙΣ

```
PROGRAM POINTS_IN_CIRCLES
  IMPLICIT NONE
  INTEGER NMAX, N, I, INSIDE_CIRCLE, IN1, IN2, IN3, SUM
  PARAMETER (NMAX = 1000)
  DOUBLE PRECISION X(NMAX), Y(NMAX), X0(3), &
    Y0(3), R(3)

  WRITE(*,*) 'ΠΟΣΑ ΣΗΜΕΙΑ ΘΑ ΕΙΣΑΓΕΤΕ;'
  READ(*,*) N
  IF (N > NMAX .OR. N <= 0) STOP

  WRITE(*,*) 'ΕΙΣΑΓΕΤΕ ΤΑ ΣΗΜΕΙΑ'
  READ(*,*) (X(I), I = 1, N), (Y(I), I = 1, N)
  WRITE(*,*) 'ΕΙΣΑΓΕΤΕ ΤΟΥΣ 3 ΚΥΚΛΟΥΣ'
  READ(*,*) (X0(I), I = 1, 3), (Y0(I), I = 1, 3), &
    (R(I), I = 1, 3)
```

ΣΥΝΕΧΙΖΕΤΑΙ...

## Παράδειγμα #8: Αριθμός σημείων μέσα στο υπόλοιπο της τομής 3 κύκλων (2/2)

ΥΠΟΛΟΓΙΣΤΕΣ I - ΣΥΝΑΡΤΗΣΕΙΣ

```
SUM = 0

DO I = 1, N
  IN1=INSIDE_CIRCLE(X(I),Y(I),X0(1),Y0(1),R(1))
  IN2=INSIDE_CIRCLE(X(I),Y(I),X0(2),Y0(2),R(2))
  IN3=INSIDE_CIRCLE(X(I),Y(I),X0(3),Y0(3),R(3))
  IF (IN1+IN2+IN3 == 1) SUM = SUM + 1
END DO

WRITE(*,*) 'ΜΕΣΑ ΣΤΟΥΣ ΚΥΚΛΟΥΣ ΕΙΝΑΙ', SUM, 'ΣΗΜΕΙΑ'
```


END

23

## Παράδειγμα #9: Αριθμός σημείων μέσα στις τομές M κύκλων

Έστω ότι δίνονται N σημεία στο επίπεδο. Βρείτε πόσα από αυτά βρίσκονται μέσα σε οποιαδήποτε από τις τομές των M κύκλων (δηλαδή πόσα είναι μέσα στις σκούρες γαλάζιες περιοχές)

ΥΠΟΛΟΓΙΣΤΕΣ I - ΣΥΝΑΡΤΗΣΕΙΣ



24

## Παράδειγμα #9: Αριθμός σημείων μέσα στις τομές M κύκλων (1/2)

```
PROGRAM POINTS_IN_CIRCLES
  IMPLICIT NONE
  INTEGER NMAX, N, M, I, J, INSIDE_CIRCLE, IN, SUM
  PARAMETER (NMAX = 1000)
  DOUBLE PRECISION X(NMAX), Y(NMAX), X0(NMAX), &
    Y0(NMAX), R(NMAX)

  WRITE(*,*) 'ΠΟΣΑ ΣΗΜΕΙΑ ΘΑ ΕΙΣΑΓΕΤΕ; ΚΥΚΛΟΥΣ;'
  READ(*,*) N, M
  IF (N > NMAX .OR. M > NMAX) STOP

  WRITE(*,*) 'ΕΙΣΑΓΕΤΕ ΤΑ ΣΗΜΕΙΑ'
  READ(*,*) (X(I), I = 1, N), (Y(I), I = 1, N)
  WRITE(*,*) 'ΕΙΣΑΓΕΤΕ ΤΟΥΣ ΚΥΚΛΟΥΣ'
  READ(*,*) (X0(I), I = 1, M), (Y0(I), I = 1, M), &
    (R(I), I = 1, M)
  25
  SYNEXIZETAI...
```

ΥΠΟΛΟΓΙΣΤΕΣ I - ΣΥΝΑΡΤΗΣΕΙΣ

## Παράδειγμα #9: Αριθμός σημείων μέσα στις τομές M κύκλων (2/2)

```
SUM = 0

DO I = 1, N
  IN = 0
  DO J = 1, M
    IN = IN + INSIDE_CIRCLE(X(I), Y(I), &
      X0(J), Y0(J), R(J))
  END DO

  IF (IN >= 2) SUM = SUM + 1
END DO

WRITE(*,*) 'ΜΕΣΑ ΣΤΟΥΣ ΚΥΚΛΟΥΣ ΕΙΝΑΙ', SUM, 'ΣΗΜΕΙΑ'

END
26
```

ΥΠΟΛΟΓΙΣΤΕΣ I - ΣΥΝΑΡΤΗΣΕΙΣ

## Εισαγωγή πινάκων σε συναρτήσεις: π.χ. μέτρο δισδιάστατου διανύσματος

```
PROGRAM VECTOR2D
  IMPLICIT NONE
  DOUBLE PRECISION P(2), MAGNITUDE, M
  WRITE(*,*) 'ΔΩΣΕ ΤΙΣ ΣΥΝΙΣΤΩΣΕΣ'
  READ(*,*) P(1), P(2)

  M = MAGNITUDE(P)
  WRITE(*,*) 'ΤΟ ΜΕΤΡΟ ΤΩΝ', P(1), P(2), 'ΕΙΝΑΙ', M

END

DOUBLE PRECISION FUNCTION MAGNITUDE ( P )
  IMPLICIT NONE
  DOUBLE PRECISION P(2)
  MAGNITUDE = SQRT (P(1)**2 + P(2)**2)
  RETURN
END
27
```

ΥΠΟΛΟΓΙΣΤΕΣ I - ΣΥΝΑΡΤΗΣΕΙΣ

## Εισαγωγή πινάκων σε συναρτήσεις

- Στην λίστα μεταβλητών εισόδου μπαίνει ΜΟΝΟ το όνομα του πίνακα
- Στην δήλωση μεταβλητών εισόδου δηλώνουμε το μέγεθος του πίνακα:
- Το μέγεθος του πίνακα μπορεί να περάσει και αυτό σαν είσοδος

ΥΠΟΛΟΓΙΣΤΕΣ I - ΣΥΝΑΡΤΗΣΕΙΣ



## Παράδειγμα #10: Μέτρο διανύσματος N διαστάσεων

ΥΠΟΛΟΓΙΣΤΕΣ I - ΣΥΝΑΡΤΗΣΕΙΣ

Γράψτε συνάρτηση που να δέχεται έναν μονοδιάστατο πίνακα και τον αριθμό των ενεργών στοιχείων του και να επιστρέφει το μέτρο του

```
DOUBLE PRECISION FUNCTION MAGNITUDE_N (X, N)
  IMPLICIT NONE
  INTEGER N, I
  DOUBLE PRECISION X(N), S

  S = 0
  DO I = 1, N
    S = S + X(I)**2
  END DO

  MAGNITUDE_N = SQRT (S)

  RETURN
END
```

29

## Γιατί εισάγουμε μόνο το πλήθος των ενεργών στοιχείων και όχι την μέγιστη διάσταση του πίνακα;

ΥΠΟΛΟΓΙΣΤΕΣ I - ΣΥΝΑΡΤΗΣΕΙΣ

- Η σωστή ερώτηση είναι: όταν εισάγουμε έναν πίνακα σε μια συνάρτηση, εισάγουμε όλο τον πίνακα (δηλαδή όλα τα στοιχεία του);
- Η απάντηση είναι ΟΧΙ: κατά την είσοδο μέσα στην συνάρτηση μεταφέρεται μόνο η διεύθυνση μνήμης του πρώτου στοιχείου του πίνακα, τίποτα άλλο (κλήση κατά αναφορά). Κατόπιν η προσπάθεια γίνεται σειριακά.
- Άρα η δήλωση της μέγιστης διάστασης δεν είναι υποχρεωτική όταν έχουμε μονοδιάστατο πίνακα
- Είναι υποχρεωτική όταν έχουμε πολυδιάστατο πίνακα

30

## Παράδειγμα #11: Ελάχιστο πίνακα

ΥΠΟΛΟΓΙΣΤΕΣ I - ΣΥΝΑΡΤΗΣΕΙΣ

Γράψτε συνάρτηση που να δέχεται έναν μονοδιάστατο πίνακα και τον αριθμό των ενεργών στοιχείων του και να επιστρέφει το ελάχιστο στοιχείο

```
DOUBLE PRECISION FUNCTION XMIN (X, N)
  IMPLICIT NONE
  INTEGER N, I
  DOUBLE PRECISION X(N)

  XMIN = X(1)
  DO I = 2, N
    IF(X(I)< XMIN) XMIN = X(I)
  END DO

  RETURN
END
```

31

## Παράδειγμα #12: Μέσος όρος πίνακα

ΥΠΟΛΟΓΙΣΤΕΣ I - ΣΥΝΑΡΤΗΣΕΙΣ

Γράψτε συνάρτηση που να δέχεται έναν μονοδιάστατο πίνακα και τον αριθμό των ενεργών στοιχείων του και να επιστρέφει τον μέσο όρο

```
DOUBLE PRECISION FUNCTION AVERAGE (X, N)
  IMPLICIT NONE
  INTEGER N, I
  DOUBLE PRECISION X(N)

  AVERAGE = 0
  DO I = 1, N
    AVERAGE = AVERAGE + X(I)
  END DO
  AVERAGE=AVERAGE/N
  RETURN
END
```

32



## Παράδειγμα #13: Σφάλμα μέσου όρου

Γράψτε συνάρτηση που να δέχεται έναν μονοδιάστατο πίνακα και τον αριθμό των ενεργών στοιχείων του και να επιστρέφει το σφάλμα του μέσου όρου

- Σε ένα σύνολο μετρήσεων  $x_i$ ,  $i=1, N$ , υπολογίζουμε

μέσος όρος

$$\bar{X} = \frac{\sum_{i=1}^N X_i}{N}$$

Σφάλμα μέσου όρου

$$\sigma_x = \left[ \frac{\sum_{i=1}^N (X_i - \bar{X})^2}{N(N-1)} \right]^{1/2}$$

33

## Παράδειγμα #13: Σφάλμα μέσου όρου

$$\sigma_x = \left[ \frac{\sum_{i=1}^N (x_i - \bar{x})^2}{N(N-1)} \right]^{1/2}$$

ΥΠΟΛΟΓΙΣΤΕΣ Ι - ΣΥΝΑΡΤΗΣΕΙΣ

```

DOUBLE PRECISION FUNCTION STD (X, N)
  IMPLICIT NONE
  INTEGER N, I
  DOUBLE PRECISION X(N), XMEAN, AVERAGE

  XMEAN = AVERAGE(X, N)

  STD = 0
  DO I = 1, N
    STD = STD + (X(I) - XMEAN)**2
  END DO

  STD = SQRT( STD / (N*(N-1)) )

  RETURN
END
  
```

34

## Παράδειγμα #14 Στατιστική ανάλυση

Κατασκευάστε πρόγραμμα που δέχεται N αριθμούς και θα κάνει στατιστική ανάλυση

```

PROGRAM STATISTICS
  IMPLICIT NONE
  INTEGER NMAX, N, I
  PARAMETER(NMAX = 1000)
  DOUBLE PRECISION X(NMAX), AVERAGE, STD

  WRITE(*,*) `ΠΟΣΕΣ ΤΙΜΕΣ ΘΑ ΕΙΣΑΓΕΤΕ`
  READ(*,*) N
  IF (N > NMAX .OR. N <= 0) STOP

  WRITE(*,*) `ΕΙΣΑΓΕΤΕ ΤΙΣ ΤΙΜΕΣ`
  READ(*,*) (X(I), I = 1, N)

  WRITE(*,*) AVERAGE(X, N), `+-`, STD(X, N)

END
  
```

35

## Παράδειγμα #15: Ελάχιστο πίνακα μεταξύ N1 και N2

Γράψτε συνάρτηση που να δέχεται έναν μονοδιάστατο πίνακα και δύο ακέραιους N1 και N2, και να επιστρέφει το ελάχιστο στοιχείο μεταξύ N1 και N2

```

DOUBLE PRECISION FUNCTION XMIN (X, N1, N2)
  IMPLICIT NONE
  INTEGER N1, N2, I
  DOUBLE PRECISION X(N2)

  XMIN = X(N1)
  DO I = N1+1, N2
    IF(X(I) < XMIN) XMIN = X(I)
  END DO

  RETURN
END
  
```

36

## Παράδειγμα #16: Θέση ελαχίστου πίνακα

Γράψτε συνάρτηση που να δέχεται έναν μονοδιάστατο πίνακα και να επιστρέφει την θέση στην οποία βρίσκεται το ελάχιστο στοιχείο του

Λύση:

- Η συνάρτηση θα δέχεται έναν μονοδιάστατο πίνακα καθώς και τον αριθμό των ενεργών στοιχείων του
- Μέσα στην συνάρτηση βρίσκουμε και το ελάχιστο στοιχείο αλλά και την θέση στην οποία βρίσκεται αυτό, και γυρνάμε πίσω την θέση
- **Συνάρτηση τύπου INTEGER**

37

## Παράδειγμα #16: Θέση ελαχίστου πίνακα

```

INTEGER FUNCTION IMIN (X, N)
  IMPLICIT NONE
  INTEGER N, I
  DOUBLE PRECISION X(N), XMIN

  XMIN = X(1)
  IMIN = 1
  DO I = 2, N
    IF (X(I) < XMIN) THEN
      XMIN = X(I)
      IMIN = I
    END IF
  END DO

  RETURN
END
  
```

38

## Εισαγωγή δισδιάστατου πίνακα

- Εδώ πρέπει να εισάγουμε τουλάχιστον το μέγιστο πλήθος γραμμών του πίνακα
- ΔΕΝ μπορούμε να εισάγουμε μόνο το πλήθος των ενεργών στοιχείων όπως κάνουμε με τους μονοδιάστατους πίνακες
- Στην γενική περίπτωση με πίνακες  $M$  διαστάσεων, δηλώνουμε επακριβώς τις πρώτες  $M-1$  διαστάσεις

39

## Παράδειγμα #17: ίχνος διδιάστατου πίνακα

Γράψτε συνάρτηση που να δέχεται έναν διδιάστατο τετραγωνικό πίνακα και να επιστρέφει το ίχνος του

```

DOUBLE PRECISION FUNCTION TRACE (X, N)
  IMPLICIT NONE
  INTEGER N, I
  DOUBLE PRECISION X(N, N)

  TRACE = 0
  DO I = 1, N
    TRACE = TRACE + X(I, I)
  END DO

  RETURN
END
  
```

Εδώ το  $N$  πρέπει αναγκαστικά να είναι η μέγιστη διάσταση<sup>40</sup>

40

## Παράδειγμα #18: άθροισμα των N πρώτων στοιχείων της διαγωνίου

Γράψτε συνάρτηση που να δέχεται έναν δισδιάστατο τετραγωνικό πίνακα και να επιστρέφει το άθροισμα των N πρώτων στοιχείων της διαγωνίου του

```
DOUBLE PRECISION FUNCTION TRACE (X, NMAX, N)
  IMPLICIT NONE
  INTEGER NMAX, I
  DOUBLE PRECISION X(NMAX, N)

  TRACE = 0
  DO I = 1, N
    TRACE = TRACE + X(I, I)
  END DO

  RETURN
END
```

41

Εδώ πρέπει αναγκαστικά να εισάγουμε και τη μέγιστη διάσταση

## Γιατί αυτή η διαφορά μεταξύ μονοδιάστατων και πολυδιάστατων;

έστω ένας πίνακας X(9), και θέλουμε να επεξεργαστούμε τα 4 πρώτα στοιχεία του

ο πίνακας X(9), όπως είναι στο κυρίως πρόγραμμα

$$\begin{pmatrix} 1 \\ 2 \\ 3 \\ 4 \\ 5 \\ 6 \\ 7 \\ 8 \\ 9 \end{pmatrix}$$

εαν τον περάσουμε σε συνάρτηση και καλέσουμε τα 4 πρώτα στοιχεία του, θα έχουμε

$$\begin{pmatrix} 1 \\ 2 \\ 3 \\ 4 \end{pmatrix}$$

42

## Γιατί αυτή η διαφορά μεταξύ μονοδιάστατων και πολυδιάστατων;

έστω τώρα ένας πίνακας X(3,3), και θέλουμε να επεξεργαστούμε το πρώτο (2,2) τεταρτημόριό του

ο πίνακας X(3,3), όπως είναι στο κυρίως πρόγραμμα

$$\begin{pmatrix} 11 & 12 & 13 \\ 21 & 22 & 23 \\ 31 & 32 & 33 \end{pmatrix}$$

ο πίνακας X, όπως είναι στη μνήμη

$$\begin{pmatrix} 11 \\ 21 \\ 31 \\ 12 \\ 22 \\ 32 \\ 13 \\ 23 \\ 33 \end{pmatrix}$$

στην συνάρτηση, εαν τον δηλώσουμε ως 2X2 πίνακα, να τι θα πάρουμε

$$\begin{pmatrix} 11 & 31 \\ 21 & 12 \end{pmatrix} \text{ λάθος}$$

εαν όμως τον δηλώσουμε ως 3X2, θα πάρουμε

$$\begin{pmatrix} 11 & 12 \\ 21 & 22 \\ 31 & 32 \end{pmatrix} \text{ σωστό!!}$$

43